

## Instantons and self-duality

This simple notebook shows a simple calculation related to the 't Hooft ansatz for instanton solutions in Yang-Mills theory. It will show you how to compute the constraint on the function  $H(r)$  which appears in this ansatz, by imposing self-duality of the field strength.

It mainly illustrates the way in which partial derivatives and substitution commands work.

```
{i,j,k,l,m,n,q,r#}::Indices(vector).
{i,j,k,l,m,n,q,r#}::Integer(1..4).
{a,b,c,d,e,f,g}::Indices(group).
{a,b,c,d,e,f,g}::Integer(1..3).
\eta^{a}_{i j}::TableauSymmetry(shape={1,1},indices={1,2}).
\partial{#}::PartialDerivative.
\epsilon^{a b c}::EpsilonTensor.
\epsilon_{i j k l}::EpsilonTensor.
\log{H}::Depends(\partial).
\delta_{i j}::KroneckerDelta.
```

Assigning list property Indices to i, j, k, l, m, n, q, r#.  
Assigning property Integer to i, j, k, l, m, n, q, r#.  
Assigning list property Indices to a, b, c, d, e, f, g.  
Assigning property Integer to a, b, c, d, e, f, g.  
Assigning property TableauSymmetry to \eta.  
Assigning property PartialDerivative to \partial.  
Assigning property EpsilonTensor to \epsilon.  
Assigning property EpsilonTensor to \epsilon.  
Assigning property Depends to \log.  
Assigning property KroneckerDelta to \delta.

First define the field strength in terms of the gauge potential:

```
F:=\partial_{i}{A_{j}^{a}} - \partial_{j}{A_{i}^{a}}
      + \epsilon^{a b c} A_{i}^{b} A_{j}^{c};
```

$$F := \partial_i A_j^a - \partial_j A_i^a + \epsilon^{abc} A_i^b A_j^c;$$

Now insert the 't Hooft ansatz in terms of the 't Hooft symbols  $\eta_{ij}^a$ :

```
@substitute!(%)( A_{i}^{a} -> \eta^{a}_{i j} \partial_{j}{\log{H}} );
```

$$F := \partial_i(\eta^a_{jk} \partial_k \log H) - \partial_j(\eta^a_{ik} \partial_k \log H) + \epsilon^{abc} \eta^b_{ik} \partial_k \log H \eta^c_{jl} \partial_l \log H;$$

```
@prodrule!(%):
@unwrap!(%);
```

$$F := \eta^a_{jk} \partial_{ik} \log H - \eta^a_{ik} \partial_{jk} \log H + \epsilon^{abc} \eta^b_{ik} \partial_k \log H \eta^c_{jl} \partial_l \log H;$$

The 't Hooft symbols  $\eta_{ij}^a$  satisfy several rules which can be used to simplify products of them (see e.g. hep-th/0012016 and Rajaraman's book):

```

etarules:= { \eta^{a}_{i j} \eta^{a}_{k l} ->
              \delta_{i k} \delta_{j l} - \delta_{i l} \delta_{k j}
              + \epsilon_{i j k l},
\epsilon^{a b c} \eta^{b}_{i k} \eta^{c}_{j l} ->
              -\delta_{i j} \eta^{a}_{k l} - \delta_{k l} \eta^{a}_{i j}
              + \delta_{i l} \eta^{a}_{k j} + \delta_{k j} \eta^{a}_{i l},
\eta^{a}_{i k} \eta^{b}_{k j} ->
              -\delta^{a b} \delta_{i j} + \epsilon^{a b c} \eta^{c}_{i j},
\epsilon_{i j l k} \eta^{a}_{l m} ->
              - \delta_{i m} \eta^{a}_{j k} - \delta_{j m} \eta^{a}_{k i}
              - \delta_{k m} \eta^{a}_{i j},
\epsilon_{i j k l} \eta^{a}_{k l} ->
              - 2 \eta^{a}_{i j} };

```

```

etarules := { \eta^a_{ij} \eta^a_{kl} \to (\delta_{ik} \delta_{jl} - \delta_{il} \delta_{kj} + \epsilon_{ijkl}), \quad \epsilon^{abc} \eta^b_{ik} \eta^c_{jl}
\to (-\delta_{ij} \eta^a_{kl} - \delta_{kl} \eta^a_{ij} + \delta_{il} \eta^a_{kj} + \delta_{kj} \eta^a_{il}), \quad \eta^a_{ik} \eta^b_{kj} \to (-\delta^{ab} \delta_{ij} + \epsilon^{abc} \eta^c_{ij}), \quad \epsilon_{ijkl} \eta^a_{lm}
\to (-\delta_{im} \eta^a_{jk} - \delta_{jm} \eta^a_{ki} - \delta_{km} \eta^a_{ij}), \quad \epsilon_{ijkl} \eta^a_{kl} \to (-2) \eta^a_{ij};

```

```

@substitute!(F)( @(etarules) ):
@distribute!(%):
@eliminate_kr!(%):
@prodsort!(%):
@canonicalise!(%):
@rename_dummies!(%);

```

$$\begin{aligned}
F &:= \eta^a_{jk} \partial_{ik} \log H - \eta^a_{ik} \partial_{jk} \log H - \eta^a_{ij} \partial_k \log H \partial_k \log H \\
&\quad - \eta^a_{jk} \partial_i \log H \partial_k \log H + \eta^a_{ik} \partial_j \log H \partial_k \log H;
\end{aligned}$$

Let us now define the dual field strength:

```

Fd := 1/2 \epsilon_{i j k l} F_{k l};

```

$$Fd := \frac{1}{2} \epsilon_{ijkl} F_{kl};$$

```

@substitute!(Fd)( F_{i j} -> @(F) ):
@distribute!(%);

```

$$\begin{aligned}
Fd &:= \frac{1}{2} \epsilon_{ijkl} \eta^a_{lm} \partial_{km} \log H - \frac{1}{2} \epsilon_{ijkl} \eta^a_{km} \partial_{lm} \log H - \frac{1}{2} \epsilon_{ijkl} \eta^a_{kl} \partial_m \log H \partial_m \log H \\
&\quad - \frac{1}{2} \epsilon_{ijkl} \eta^a_{lm} \partial_k \log H \partial_m \log H + \frac{1}{2} \epsilon_{ijkl} \eta^a_{km} \partial_l \log H \partial_m \log H;
\end{aligned}$$

```

@canonicalise!(%):
@substitute!!( % )( @(etarules) ):
@distribute!(%):
@eliminate_kr!(%);

```

$$\begin{aligned}
Fd := & \frac{1}{2} \eta^a_{jl} \partial_l \log H + \frac{1}{2} \eta^a_{li} \partial_l \log H + \frac{1}{2} \eta^a_{ij} \partial_{mm} \log H + \frac{1}{2} \eta^a_{jl} \partial_l \log H \\
& + \frac{1}{2} \eta^a_{li} \partial_l \log H + \frac{1}{2} \eta^a_{ij} \partial_{mm} \log H + \frac{1}{2} \eta^a_{ji} \partial_m \log H \partial_m \log H \\
& + \frac{1}{2} \eta^a_{ji} \partial_m \log H \partial_m \log H + 2 \eta^a_{ij} \partial_m \log H \partial_m \log H \\
& - \frac{1}{2} \eta^a_{jl} \partial_l \log H \partial_i \log H - \frac{1}{2} \eta^a_{li} \partial_l \log H \partial_j \log H - \frac{1}{2} \eta^a_{ij} \partial_m \log H \partial_m \log H \\
& - \frac{1}{2} \eta^a_{jl} \partial_l \log H \partial_i \log H - \frac{1}{2} \eta^a_{li} \partial_l \log H \partial_j \log H - \frac{1}{2} \eta^a_{ij} \partial_m \log H \partial_m \log H;
\end{aligned}$$

```

@prodsort! (%):
@canonicalise! (%):
@rename_dummies! (%):
@collect_terms! (%);

```

$$Fd := \eta^a_{jk} \partial_{ik} \log H - \eta^a_{ik} \partial_{jk} \log H + \eta^a_{ij} \partial_{kk} \log H - \eta^a_{jk} \partial_i \log H \partial_k \log H + \eta^a_{ik} \partial_j \log H \partial_k \log H;$$

The condition that the configuration is self-dual, i.e.  $F = *F$ , now reduces to

```

eq:= @(F) - @(Fd):
@collect_terms! (%);

```

$$eq := -\eta^a_{ij} \partial_k \log H \partial_k \log H - \eta^a_{ij} \partial_{kk} \log H;$$

This is precisely the condition found in Rajaraman's book, equation (4.60b).